

Intusoft Newsletter

Personal Computer Circuit Design Tools

September 1993 Issue



Copyright © Intusoft, All Rights Reserved

(310) 833-0710 Fax (310) 833-9658

Does Your Simulator Stack Up?

Does your SPICE simulator measure up? IsSPICE3 is based on the latest U.C. Berkeley code, SPICE 3F.2, with additional models and simulation algorithms from various universities and consultants around the world.


Contrary to what many non-SPICE simulation vendors would have you believe, SPICE development is not stagnant. While the core algorithms have been tested over time, many new features have been recently added to IsSPICE3 in several areas, for instance:

New Built-in Models: Behavioral Modeling (in-line equations, If-Then-Else), switches, enhanced JFET/MESFET models, BSIM1/2, Level 6 (MOSFET discontinuities removed), lossy t-lines
New Analyses: Simulation scripts, Simulation Breakpoints
New Outputs: Computed Device Parameters, Print Expressions
New Convergence Algorithms: Gmin Stepping

In This Issue

- 2 Power Schottky/
Soft Recovery Diodes
- 9 Does Your SPICE
Stack Up?
- 10 PZ & AC/DC Sensitivity
- 12 The Modeling Corner:
AD & Comlinear ICs
Making AM Signals
- 13 Analog Mixer Circuit
Intusoft NEWS

continued on page 32-9





Power Schottky/Soft Recovery Diodes

Schottky rectifiers are widely used in switched mode converters due to their inherently lower forward voltage characteristics and superior recovery time. Schottky diodes can normally be modeled using the built-in SPICE diode. However, in the case of power Schottky rectifiers, two effects must be accounted for that are not part of the built-in SPICE model. The first is the conductivity modulation in the forward conductance and the second is the reverse leakage current which is an exponential function of both temperature and reverse voltage. $IsSPICE3$ contains the functions necessary to construct a reasonable power Schottky model that adds these enhancements.

Figure 2 shows the subcircuit schematic of the new power Schottky rectifier. The diode, DIO, accounts for most of the traditional Schottky behavior. A Schottky diode is a majority carrier device whose reverse recovery time is zero. However, its larger junction capacitance, CJO, will produce a similar effect. Therefore, the minority carrier storage time parameter, TT, is left at its default value of 0. The emission coefficient, N, is normally set close to 1. The saturation current value, IS, is typically much higher (two orders of magnitude) than for p-n diodes. The values for saturation current temperature exponent, XTI =2, and energy gap, EG=0.69, are set based on the values expected for Schottky barrier diodes. In comparison to the manufacturer's reverse current data, the high temperature leakage appears to be slightly overstated by these values, but within likely product variations.

The voltage sensitive effects are produced by BDIO, an exponential nonlinear current source. Such a source would have been considerably more difficult to construct with SPICE 2G.6 based programs. With the in-line equation feature of $IsSPICE3$,





this function becomes trivial. The forward recovery of the diode, not readily verifiable because of the lack of data, is modeled by the time constant of RFR and CFR. Deleting CFR and shorting RFR makes the forward recovery instantaneous.

Figures 3 and 4 show the VF vs. IF response in the forward and reverse directions over temperature. All the waveforms were obtained in one simulation pass using the simulation scripting features of IsPICE3. The netlist for the IR10CTQ150 (10A 150V) part used here is listed in Table 1 on pg. 32-8. Several other models are enclosed on the subscription floppy disk.

Modeling Soft Recovery

Modeling of the transient behavior of power diodes is extremely important for power electronics. By properly choosing the optimum diode recovery characteristic, the designer can significantly reduce switching losses, voltage spikes, RFI and EMI. As




Figure 4, Graph of the reverse leakage current vs. temperature shows that the current is an exponential function of both temperature and voltage.

discussed earlier, the built-in diode model in SPICE does not exhibit forward recovery. In addition, the default diode model exhibits an abrupt “snappy” reverse recovery. With the help of IsSPICE3 we will look at how to rectify the situation by enhancing the built-in model. The subcircuit presented here is based on a simplified physical model for a high voltage p-i-n structure operating in high level injection as is typical for most power diodes.

A variety of approximating models are possible for improving the accuracy of simulation without resorting to the use of C-code subroutines or other proprietary modeling languages. Both of these approaches suffer from a myriad of syntax, compatibility and logistical hurdles. One such model has been proposed in [1]. This model can easily be implemented with IsSPICE3 or even SPICE 2, though somewhat less efficiently.

Reverse Recovery

SPICE’s internal recovery model of a diode only models charge storage as a function of the diode current. When the diode is commutated, the diode will remain on until all the charge is removed, after which, it will suddenly snap off. The reverse recovery time, T_{rr} , consists of two components T_a and T_b . T_a is caused by the charge storage in the depletion region of the junction and represents the time between the zero current crossing and the peak reverse current, I_{rrm} . T_b is caused by the charge storage in the bulk semiconductor material. Turn off proceeds with an ultimate reversal of the biasing current and the junction remaining forward biased (charged). This continues until the charge is removed from one of the junctions adjacent to the -i layer. The -i layer is still charged as the reverse voltage rapidly increases.





Figure 5, Curves showing the diode voltage, instantaneous power (Dpower), and energy (integral of Dpower) curves as the new diode subcircuit is turned off and on.

Such behavior is appropriate for small signal low-voltage diodes, but not for power diodes which are typically formed with a lightly doped layer between the p-n junctions to withstand high reverse voltages. No additional diode state variables are available in SPICE to represent the physical charge storage in the lightly doped internal regions of such a p-i-n structure. As a consequence, spurious oscillations are seen in SPICE simulations that attempt to use the standard diode model with non-zero values of TT. Not only can the simulation be inaccurate, but it can often refuse to converge. Adding to the confusion is the fact that the oscillation amplitude and duration can vary depending on the .TRAN TSTEP and RELTOL values chosen.

The enhanced diode subcircuit, DBEHAV, is shown in Table 1. The diode DMODEL models the space charge capacitance. Other characteristics are removed by making RS and TT=0 and IS small. The standard diode behavior and reverse/forward recovery effects are added using the in-line equation feature (B element) of ISSPICE3. The parameter IS1 is similar to the IS diode model parameter. ISE and the RS device value are chosen for the best high current performance. N, shown in the BE element's equation ($2 \cdot VTA$), is set to 2 for high level injection behavior, but can be set to 1 to function as a low-voltage p-n junction diode. Tm (transit time) and Tau (carrier lifetime) are determined from a diode turn-off current waveform. As TM increases, stored charge is removed more slowly from the base region. This has the effect of making Ta longer and Irrm larger. RMO controls the forward recovery.


High diode power dissipation often occurs during the final current tail interval as evidenced by the power dissipation and



Reverse Recovery *continued*

energy curves in Figure 5. The power dissipation curve was created by the IsSPICE3 Print Expression “alias dpower (v(3) - v(2)) * i(vsense)”. The power was then integrated in INTUSCOPE to get the energy. Figure 6 shows the diode current for the subcircuit diode (WFM.1) and for a comparable SPICE diode .model statement (.MODEL DMOD D IS=20F RS=4M N=1 CJO=100P M=.5 TT=100N, WFM.2). The Qrr charge curve from the new diode is shown as waveform 3. As shown in Figure 6, soft recovery diodes exhibit a faster recovery time and a lower value of I_{rrm} than conventional diodes.

Forward Recovery



During the turn-on transient, a high forward voltage builds up across the diode because of the low conductivity in the -i region. As the injected carrier concentration increases, the voltage across the -i region soon decreases to a normal steady state diode forward drop. This effect gives rise to the initial voltage overshoot in the turn-on forward voltage as the -i region must become flooded with current carriers for its full conductivity to be effective. The forward voltage peak, as shown to the left and in Figure 5, is 10.197V. Waveform 2 shows the standard SPICE

forward recovery.


Simulating an IGBT Chopper Circuit

Figure 7 contains a simple IGBT chopper circuit using the new diode model. To simplify the circuit, the gate drive circuitry has been replaced by a piece-wise linear voltage source, V4. To study the diode behavior, the IGBT is turned on to generate current flow in inductor, L1, then turned off to cause current to flow into the diode (DBEHAV), then back on again to snap the diode off. The resulting diode turn off is shown in Figure 8.

Gear Integration Speeds Power Simulations

IsSPICE3 includes a very valuable option, preferred by power engineers, to change to the Gear method of integration. Gear integration tends to find a stable time domain solution more quickly than the default trapezoidal method for many types of circuits. This results in a simulation that spends less time in high frequency calculations and is less prone to artificial numerical oscillations. SPICE simulators that do not have Gear integra-

Figure 7, IGBT chopper circuit using the enhanced diode model along with an Intusoft IGBT model.



tion can be more prone to convergence problems and often simulate more slowly than IsSPICE3 for switching power circuits. Figure 8 shows the diode response using the Gear and trapezoidal integration methods.

SPICE = AHDL

Lauritzen and Ma incorrectly state in their paper [1] that macro models are

normally valid only over a very narrow range of circuit operating conditions. The variety of models and verified simulation results presented in past newsletters is evidence that macro models are far more practical and robust than indicated. SPICE macro-models are the AHDL of the past and future. Macro modeling gives the user several advantages over other proposed methods; devices can be put together in a hierarchical manner, with pre-defined blocks whose convergence and behavior is well established and well understood by the user. And when the Berkeley syntax is adhered to, such as in the case of all Intusoft models, the models are portable between any vendor's SPICE simulator. Clearly, the new behavioral modeling (If-Then-Else and in-line equations) features of IsSPICE3 are easy to incorporate into a model and contain virtually all of the AHDL elements often used by so-called "mathematical simulators". These features allow the user to bridge the gap between modeling with detailed physical based equations and pre-defined macro blocks in order to have the best of both modeling worlds.




Figure 8, Turn-off response for the diode, Dbehav, in the IGBT chopper circuit, shows the different recovery and superior performance of Gear integration.

Table 1, The enhanced SPICE diode with forward/reverse recovery and the power Schottky rectifier model.

```
.SUBCKT DBEHAV 1 9 {IS1=1E-6 TM=100N TAU=100N RMO=1 VTA=.0259 CAP=100P ISE=1E-30}
* Connections      A C
* Following components include for space charge capacitance
DMODEL 1 2 DCAP
.MODEL DCAP D (IS=1E-21 RS=0 TT=0 CJO=(CAP)) ; M/VJ can also be added
BD 1 2 I=V(5,6) / (TM) + {ISE} * (E^(V(1,2) / {VTA}) - 1) ; Diode conductive currents
* Following components model reverse recovery
BE 5 0 V={IS1} * {TAU} * (E^(V(1,2) / (2*{VTA})) - 1) ; V(5)=QE, N = 2 in (2* {VTA})
*RE 5 0 1E6 ; QE = Junction charge
BM 6 0 V=(V(5) / (TM) - I(VSENSE1)) * ((TM) * {TAU} / ((TM) + {TAU})) ; V(6)=QM
*RM 6 0 1E6 ; QM = Charge in the base region
BDM 7 0 V=V(6)
VSENSE1 7 8 ; I(Vsense1)=dQM/dt
CDM 8 0 1
RDM 8 0 1E9
* Models high current effects and forward recovery
RS 2 3 4E-3
BMO 3 4 V=2*{VTA} * {RMO} * {TM} * I(VSENSE2) / ((V(6) * {RMO}) + ({VTA} * {TM}))
VSENSE2 4 9
.ENDS

.SUBCKT IR10CTQ150 1 2 ; 150V 10A Power Schottky Diode
*CONNECTIONS      A C
LPKG 1 3 8N ; Lead Inductance
RS 3 4 13MOHM
RB 4 5 105MOHM
RFR 3 6 1K ; Forward Recovery
CFR 5 6 10PF ; Forward Recovery
MFR 5 6 4 5 BULK
.MODEL BULK NMOS (VTO=.19 KP=320)
DIO 5 7 DMOD1
VDIO 7 2
.MODEL DMOD1 D(IS=50NA N=.9 XTI=2 EG=.69) ; Schottky Settings XTI=2, EG=.69
BDIO 5 2 I= I(VDIO) * EXP (.013 * V(2,5)) ; Voltage Sensitive Reverse Current Effect
DCAP 5 2 DMOD2
.MODEL DMOD2 D(CJO=420P M=.6 VJ=.7)
CPAR 5 2 23P
.ENDS
```

References

- [1] P.O. Lauritzen and C.L. Ma, "A simple diode model with forward and reverse recovery", IEEE Transactions on Power Electronics, Vol. 6, No. 2 Apr. 1991
- [2] Gerald Stanley, "A simple SPICE3 Model for Schottky Power Diodes", Crown International, 9-92, private correspondence
- [3] Gerald Stanley, "SPICE32 Rect(n) Functions", Crown International, 9-92, private correspondence
- [4] Ned, Mohan, "Power Electronics: Computer Simulation, Analysis, and Education", Minnesota Power Electronics Research & Education, P.O. Box 14503, Minneapolis, MN 55414, 1992
- [5] International Rectifier, "Power Modules Designer's Handbook", 1991
- [6] Y. Khersonsky, M. Robinson, D. Gutierrez, "The Hexfred™ Ultrafast Diode in Power Switching Circuits", AN-989, International Rectifier, 1992
- [7] Unitrode, "Semiconductor Databook and Application Notes", 1989
- [8] P. Antognetti and G. Massobrio, "Semiconductor Device Modeling with SPICE", 2nd Ed., McGraw-Hill, 1993

Does Your SPICE Stack Up?

continued
from page
32-1

In this article we will touch on a few of the newest features of IsSPICE3. If your using a different SPICE program that isn't based on Berkeley SPICE 3F, you'll be surprised at some of the new capabilities your SPICE program is missing.

One of the most powerful new features of IsSPICE3 is the Interactive Command Language which contains all the standard SPICE functions plus, conditional branching, breakpoints, and additional output features. Complete simulation scripts can be written, stored and included in any netlist as "canned" test procedures. For example, the circuit in Figure 1 is that of a two-stage CMOS op-amp. The simulation was performed with a simple script written to run a DC analysis and then several AC and Transient analyses over temperature. The Mosfet gm and gds values, available in IsSPICE3, were used as follows:

Portion of the simulation script used to generate Figure 1

```

TRAN 1M 200M ; Run a transient analysis
LET A1= -1 * @M1[GM] / (@M2[GDS] + @M4[GDS]) ; Create the stage 1 gain vector
LET A2= -1 * @M6[GM] / (@M6[GDS] + @M7[GDS]) ; Create the stage 2 gain vector
LET A0=(A1 * A2) ; Create the DC open loop gain vector
PRINT A1 A2 A0 @M1[ID] @M7[IB] ; Print the gain vectors, currents (drain/bulk)
PRINT @M2[GM] @M6[P] @R1[P] ; transconductance, & power dissipation
    
```

A0 variations are shown in the Tran graphs in Figure 1. Note: SPICE 2 based programs only allow global circuit temperature variations. They do not allow the temperature to be changed independently on a single device, like IsSPICE3, and do not give access to device parameters such as gm and gds during the transient analysis. Also, IsSPICE3 allows all aspects of the circuit to be altered between the multiple analyses in a script (as compared to the limited capability of the .STEP function found in some popular SPICE programs).

```

* Elliptic (Cauer) - lowpass filter
* FilterMaster Professional
C2 2 0 6.198E-10
L3 2 3 7.605E-07
C3 2 3 7.190E-11
C4 3 0 7.658E-10
L5 3 100 5.921E-07
C5 3 100 2.013E-10
C6 100 0 5.245E-10
VIN 1 0 AC 2 0
R1 1 2 5.000E+01
R7 100 0 5.000E+01
.PZ 1 0 100 0 VOL PZ
.PRINT AC V(100) VP(100)
.END
    
```

```

----- Pole - Zero Analysis -----
Transfer function of V(100) / V(1) in Volts / Volt
Poles Real Imag Number
-----
-2.18573e+07 0.000000e+00 1
-1.42716e+07 4.366427e+07 1
-1.42716e+07 -4.36643e+07 1
-3.93393e+06 6.258026e+07 1
-3.93393e+06 -6.25803e+07 1
Zeros Real Imag Number
-----
1.618602e-24 1.352340e+08 1
1.618602e-24 -1.35234e+08 1
0.000000e+00 9.159686e+07 1
0.000000e+00 -9.15969e+07 1
    
```

Figure 9, The output from an IsSPICE3 pole/zero analysis of the lowpass filter created by FILTERMASTER (left) shows the real and imaginary roots (right).

Pole-Zero and AC-DC Sensitivity Analyses

A new analysis introduced in Berkeley SPICE 3E, and incorporated into IsSPICE3, is the pole-zero analysis. The circuit in Figure 9 is a low pass filter designed by FILTERMASTER, a filter synthesis program. While FILTERMASTER performs a similar function, IsSPICE3 can be used to find the poles and zeros of an AC transfer function between any two points in both passive and active circuits.

Most SPICE programs only have a DC sensitivity analysis. IsSPICE3 adds an AC sensitivity analysis as well. To see how these analyses can be used to perform a worst case simulation, we will look at the active low pass filter in Figure 10. The simulation script, shown in Figure 10, performs the entire simulation in one pass saving all the results in the output file. First, a DC sensitivity is performed to find which direction the

```

Simulation control script using IsSPICE3's Interactive Command Language
.control                                     ; begin the script
op                                           ; operating point
let rtol=0.1                               ; set resistor tolerance to 10%
function tol(r,s) r + r * ((s>0) * rtol-(s<=0) * rtol) ; create tolerance function
print v(4)                                  ; print nominal output voltage
show r                                       ; show the resistor models/values
sens v(4)                                   ; run a DC sensitivity analysis
print r1 r2 r3                             ; print the resistor sensitivities
alter @r1[resistance] = tol(@r1[resistance],r1) ; change the resistor values to...
alter @r2[resistance] = tol(@r2[resistance],r2) ; ... their max/min values based...
alter @r3[resistance] = tol(@r3[resistance],r3) ; ... on the sensitivity results
op                                           ; operating point
echo worst case                             ; write into the output file
print v(4)                                  ; Worst Case Output Voltage
show r                                       ; show the resistor models/values
sens v(4)                                   ; rerun the sensitivity analysis
print r1 r2 r3                             ; show the new sensitivities
sens V(4) ac dec 10 100 100k              ; run an AC sensitivity analysis
print norm(mag(c1)) norm(mag(c2))         ; print cap. sensitivities vs. Freq.
.endc                                        ; end the script

```

Simulation Results
Initial starting v(4) = -8.40353e-01
Initial resistances r3=1e+04, r2=1e+04, r1=1e+04
Sensitivities r1=-9.76087e-06, r2=-9.76087e-06, r3= 2.027591e-05
Worst case v(4) = -8.03726e-01
Final resistances r3=1.1e+04, r2=9e+03, r1=9e+03




Figure 10. The IsSPICE3 simulation script performs a worst case analysis, finding the output voltage at node 4 and the corresponding resistor values that produce it. An AC sensitivity analysis is also run. The entire script can be stored away in a file (SENS.HDR) and included in the SPICE netlist (.include sens.hdr) as a standard test procedure.




Figure 11, The AC sensitivity shows at what frequency the capacitor values are most sensitive with respect to the chosen output voltage, V(4).

component values should be moved in order to give the worst case maximum output voltage. The tolerance (tol(r,s)) and the alter functions are used to move the resistor values to their maximum or minimum depending on the magnitude of the sensitivity results. The worst case output voltage is then printed along with the final component values. Next, an AC sensitivity analysis is performed in order to find the frequency at which the capacitor values are most sensitive (Figure 11).

Speed... and Models....

The iterative nature of circuit design makes the simulator's speed a very important factor. IsSPICE3 is a 32-bit program written in C/C++. Several compilers were evaluated before selecting the best one. These factors make IsSPICE3 one of the fastest SPICE programs in the world. Check your SPICE version to see if it is using 16-bit code. If so, you are only getting half the speed of IsSPICE3. The speed of IsSPICE3 translates into immediate productivity gains by allowing you to do more in a shorter amount of time. Send us your SPICE netlist. The Intusoft technical support staff will be happy to show you how fast it runs on IsSPICE3.


Models are one of the most important parts of a simulation system. Without them, and the proper support, you can't simulate. The model libraries in the new ICAP/4DOS system have been updated with over 1500 new models, bringing the total to over 3000. New models include: GaAs diodes, 3 phase bridges, Darlington and power BJTs, power Mosfets, SCRs, Triacs, IGBTs, pentodes and triodes, digital ICs (ABT, FAST, ALS, HC/HCT), PWM ICs, analog switches, and coaxial cables. Many of the models are not available from any other vendor.

Modeling Corner

In this issue of The Intusoft Modeling Corner we bring you the latest models from Analog Devices and Comlinear. The newsletter floppy disk contains 40 new models provided by Analog Devices for a variety of new ICs including several analog switches (ADG411-ADG413). Comlinear has also added to their SPICE offering and their library now includes over 20 models. In addition to voltage and current feedback op-amps, Comlinear has made models for the CLC520 (AGC+Amp), CLC522 (Multiplier), and CLC532 (Multiplexor). The new models are included on the newsletter floppy disk for subscribers.


Stimulating Circuits

The signal generators in IsSPICE3 cover a wide variety of conditions. However, it may not be obvious to the novice user that virtually any signal can be constructed by combining the standard generators with other basic elements. For instance, IsSPICE3 contains an FM signal generator, but not an AM generator. To create the AM source we can combine two SIN generators with an E (VCVS) multiplier as shown in Figure 12. The subcircuit is parameterized, allowing it to function in a manner similar to other "built-in" IsSPICE3 signal generators.



Comlinear Models an Analog Mixer

A double balanced mixer normally works by multiplying an RF input with a LO input through the use of a diode bridge. The same result can be obtained with the circuit in Figure 13. The circuit's advantages include: The LO input is ECL and TTL compatible, excellent isolation is maintained between all ports, and the performance is good at low frequencies. The AM source (11MegHz with 0.5MegHz modulation) is used as the RF input while a pulse source (10MegHz) is used for the LO input. The frequency spectrum reveals the fundamental and sideband harmonics.



Intusoft NEWS: CompuServe Information Service

Intusoft hosts section 1 of the CADD/CAM/CAE Vendor forum on CompuServe. A variety of technical articles, application notes, demos and free SPICE models are posted there. To connect with Intusoft you can navigate through CompuServe's computing support menus to reach the CADD/CAM/CAE Vendor forum or type "Go CADDVEN" at any ! point prompt. Then select the "All CADD/CAM/CAE" section. Internet users can also send e-mail messages to Intusoft at info@intusoft.com".

EDN Article Focuses on Power Electronics

In the August 19, 1993 EDN magazine, an article written by Intusoft entitled "Improved Spice Model Simulates Transformer's Physical Processes" appeared. The article discusses how to model transformers and nonlinear magnetics in SPICE. Check it out!